
Orientation

REF: For an orientation to this manual, please refer to the “Orientation” section 
in the Kernel 8.0 Developer's Guide: Main Directory User Guide.

https://www.va.gov/vdl/documents/Infrastructure/Kernel/krn_8_0_dg.pdf#Main_Orientation

1 Introduction 
The Kernel 8.0 Developer’s Guide: Error Processing Developer Tools User Guide is part of 
the Kernel 8.0 and Kernel Toolkit 7.3 Developer’s Guide. 

2 Direct Mode Utilities 
These Error Processing Direct Mode Utilities can be run from Programmer mode. They 
are not, however, APIs and Extrinsic Functions; instead, they are provided for 
convenience.

2.1 >D ^XTER 
You can call the ^XTER direct mode utility from Programmer mode. It is the same as 
using the Error Trap Display [XUERTRAP] option.

2.2 >D ^XTERPUR 
You can call the ^XTERPUR direct mode utility from Programmer mode. It is the same 
as using the Clean Error Trap [XUERTRP CLEAN] option.

3 Application Programming Interfaces ﴾APIs﴿ 
Several Error Processing APIs and Extrinsic Functions are available for developers. These 
APIs and Extrinsic Functions are described below.

3.1 $$EC^%ZOSV: Get Error Code 
Reference Type: Supported

Category: Operating System Interface

ICR #: 10097

Description: The $$EC^%ZOSV extrinsic function returns the most recent error 
message recorded by the operating system.

Format: $$EC^%ZOSV

Input Parameters: none.

Output: returns: Returns the most recent error code/message.

3.1.1 Example 
>S X=$$EC^%ZOSV 

 

3.2 ^%ZTER: Kernel Standard Error Recording Routine 
Reference Type: Supported

Category: Error Processing

ICR #: 1621

Description: Kernel sets the Error Trap in ZU so that all user errors are trapped. 
In this context, when an error occurs, the optional %ZT input array 
is set to indicate the user’s location in the menu system. Then, call 
^%ZTER to record this information in the ERROR LOG (#3.075) file.

The application-specific Error Trap routine, when it is called because 
of an error, can then use the ^%ZTER API to record error 
information in the ERROR LOG (#3.075) file if it decides that it needs 
to. ^%ZTER gathers all available information such as local symbols 
and last global reference and stores that information in an entry in 
the ERROR LOG (#3.075) file.

The simple example below shows an application that replaces the 
standard Kernel Error Trap with its own Error Trap. When an error 

occurs, and the application’s Error Trap routine is called, it calls 
$$EC^%ZOSV to see what type of error occurred. If an end-of-file 
(EOF) error occurs, it lets the application continue. Otherwise, it calls 
^%ZTER to record the error and then quits to terminate the 
application.

NOTE: The recording mechanism of ^%ZTER also functions in 
the absence of an error. In a debug mode, this would enable a 
developer to record local symbols and global structures at 
predetermined places within code execution for later checking.

NOTE: As of Kernel Patch XU*8.0*431, the ^%ZTER error trap 
routine checks a count (limit) in the ERROR TRAP SUMMARY 
(#3.077) file and stops recording errors once this limit has 
been reached. This limit is initialized to 10 but can be changed 
by the sites. To change the value, use VA FileMan to edit the 
ERROR LIMIT (#520.1) field in the KERNEL SYSTEM 
PARAMETERS (#8989.3) file.

Format: ^%ZTER

Make sure to perform the following steps before calling this API:

1. NEW all non-namespaced variables.

2. Set all input variables.

3. Call the API.

Input Variables: %ZT: (optional) The %ZT array can be used to identify 
a global node whose descendants should be 
recorded in the error log. When called within the 
standard Kernel Error Trap, %ZT is set to record 
the user’s location in the menu system:

>S %ZT("^TMP($J)")="" 
>D ^%ZTER

Output Variables: %ZTERROR: Calls to the error recorder always return this 
variable. It has the error name and error type as 
its first and second caret-delimited (^) pieces, for 
example, %ZTERROR=UNDEF^P. While the first 

piece is always defined since it is retrieved from 
the operating system, the second piece could be 
missing if unavailable from the ERROR 
MESSAGES (#3.076) file.

3.2.1 Examples 

3.2.1.1 Example 1 

Figure 1 is an example of the Error Trap:

Figure 1: Error Trap—Example

ZXGP ; 999/NV - sample routine ; 23-FEB-95
         ;;1.0;;
         ;
FILEOPEN ;
         ;
         ; This code resets the error trap routine that is stepped to 
         ; when an error occurs.
         ;
         N $ESTACK,$ETRAP S $ETRAP="D ERR^ZXGP"
         ;
         ; Open a file, and read lines from it until End-of-file (EOF)
         ; is reached.
         ;
         K %ZIS S %ZIS=""
         S %ZIS("HFSNAME")="MYFILE.DAT",%ZIS("HFSMODE")="RW"
         D ^%ZIS Q:POP
         F  U IO R LINE:DTIME U IO(0) W !,LINE
         ;
FILECLOS ;
         ;
         D ^%ZISC Q
         ;
ERR      ;
         ; This is the application specific error trap.
         ;
         I $$EC^%ZOSV["ENDOFILE" S $ECODE="" G FILECLOS ; continue if EOF error
         D ^%ZTER ; record the error if anything other than EOF
         D UNWIND^%ZTER ; unwind the stack, return to caller.
         Q
         ;

3.2.1.2 Example 2 

To test the error limit set in the ERROR LIMIT (#520.1) field in the KERNEL SYSTEM 
PARAMETERS (#8989.3) file, run the following:

>F I=1:1:20 D APPERROR^%ZTER("My Application Error")

Check the error trap and see how many errors with "My Application Error" get recorded 
in the Kernel error log (that is ERROR LOG [#3.075] file). If the value in the ERROR 
LIMIT (#520.1) field is set to 10, there should just be 10 occurrences of the “My Error” 
error in the Kernel error log.

NOTE: For more information, see the “APPERROR^%ZTER” API.



Kernel 8.0 Developer’s Guide: 6 August 2025
Error Processing Developer Tools User Guide

3.3 APPERR^%ZTER: Set Application Error Name in Kernel 
Error Trap Log 

Reference Type: Supported

Category: Error Processing

ICR #: 1621

Description: The APPERR^%ZTER API sets the "application error" text passed in 
as the error name in the Kernel error trap log (that is ERROR LOG 
[#3.075] file).

NOTE: The APPERR^%ZTER API replaces the need to set $ZE 
before calling the ^%ZTER: Kernel Standard Error Recording 
Routine API:

Before:

>S $ZE="application error" D ^%ZTER

After:

>D APPERROR^%ZTER("application error")

NOTE: This API was released with Kernel Patch XU*8.0*431.

Format: APPERROR^%ZTER("application error")

Input Parameters: “application error”: This input parameter is the "application 
error" name that gets displayed in the Kernel 
error trap log (that is ERROR LOG [#3.075] file).

Output: returns: Displays the "application error" text passed in as 
the error name in the Kernel error trap log (that 
is ERROR LOG [#3.075] file).

3.3.1 Example 
>DO APPERROR^%ZTER("My Application Error") 

 



Kernel 8.0 Developer’s Guide: 7 August 2025
Error Processing Developer Tools User Guide

Check the Kernel error trap and see if there is an error called "My Application Error".

3.4 $$NEWERR^%ZTER: Verify Support of Standard Error 
Trapping (Obsolete) 

 NOTE: The $$NEWERR^%ZTER API is obsolete, because all VA systems support 
the standard error trapping.

Reference Type: Supported

Category: Error Processing

ICR #: 1621

Description: The $$NEWERR^%ZTER extrinsic function reports if the current 
platform supports the standard error trapping. It returns:

· 1—If the standard error trapping is supported.

· 0—For all other cases.

Format: $$NEWERR^%ZTER

Input Parameters: none.

Output: returns: Returns:
· 1—If the standard error trapping is supported.

· 0—For all other cases.

3.5 UNWIND^%ZTER: Quit Back to Calling Routine 
Reference Type: Supported

Category: Error Processing

ICR #: 1621

Description: Use the UNWIND^%ZTER API after a package Error Trap to quit 
back to the calling routine. Control returns to the level above the 
one that NEWed $ESTACK.

Format: UNWIND^%ZTER

Input Parameters: none.

Output: none.

3.5.1 Example 
Main:

Figure 2: UNWIND^%ZTER API—Main Code Example 

S X=1 D SUB
W X
Q SUB    N $ESTACK,$ETRAP S $ETR="D ERROR"
S X=1/0
Q

Usage:

Figure 3: UNWIND^%ZTER API—Usage

D ^%ZTER ;This will record the error info and clear $ECODE
S ^XXX="Incomplete record"
G UNWIND^%ZTER



Kernel 8.0 Developer’s Guide: 9 August 2025
Error Processing Developer Tools User Guide

4 Appendix A—Revision History Archive 
This section should be used to show all document revision history prior to the current 
year.

REF: For the most recent, current year document revision history, see the 
“Revision History” section.

Date Revision Description Author

TBD TBD TBD TBD

Glossary

REF: For a glossary of terms specific to Kernel, see the “Glossary” section in the 
Kernel 8.0 Developer's Guide: Main Directory User Guide.

REF: For a list of commonly used terms and acronyms, see the VA Glossary 
Power App.

https://www.va.gov/vdl/documents/Infrastructure/Kernel/krn_8_0_sm.pdf#Main_Glossary

